

INFT6303 Computer Game Production

Learning Journal

Trimester 3, 2010

Leung Yat Chun

(c3095032)

Index

Index

6 Sep. 2010

Discussion Board > 01 – My Game

Player 1 (Action)

Player 2 (Puzzle)

7 Sep. 2010

Several items learnt (Chapter 1):

Topics

The interface

ActionScript

Discussion Board > General Q&AS

Discussion Board > 01 – My Game

8 Sep. 2010

Several items learnt (Chapter 2):

Functions (setChildIndex)

Buttons

EventListener and MouseEvent

Drawing using graphics

More TextField

9 Sep. 2010

Several items learnt (Chapter 2):

Functions (string.fromCharCode(), getTimer())

Events (ENTER_FRAME, Mouse & Keyboard Events, Timer)

Properties (MouseX/Y)

More TextField

Animations

Discussion Board > 01 – My Game

10 Sep. 2010

Several items learnt (Chapter 2):

User Interaction

My handleDragDrop function

Collision Detection

External data

Discussion Board > 01 – My Game

13 Sep. 2010

Several items learnt (Chapter 2):

Custom Cursors

Playing Sounds

Loading screen (and my LoadingControl)

[Functions \(Random, floor, trunc\)](#)
[Properties \(System information, Game theft\)](#)
[Array](#)
[Display time](#)

[14 Sep, 2010](#)

[Several items learnt \(Chapter 3\):](#)

[Creating game pieces](#)
[MovieClips](#)
[More Array \(splice\) command](#)
[Events](#)
[Properties](#)
[Hard code values](#)

[15 Sep, 2010](#)

[Flip Animations](#)
[My Animation class](#)

[16 Sep, 2010](#)

[My WrapPanel class](#)

[17 Sep, 2010](#)

[Several items learnt \(Chapter 4\):](#)

[Array Functions](#)
[Dictionary](#)

[20 Sep, 2010](#)

[Several items learnt \(Chapter 4\):](#)

[Properties - buttonMode, useHandCursor, tabEnabled, mouseChildren](#)
[My Animation abstract class and FlipAnimation/MoveAnimation classes](#)

[21 Sep, 2010](#)

[My Animation.StartSequential and StartParallel static function](#)
[My GetRelativePosition function](#)

[23 Sep, 2010](#)

[Several items learnt \(Chapter 5\):](#)

[Animation class. update](#)
[My implementation of Random cards](#)
[Discussion Board > General Q&As](#)

[24 Sep, 2010](#)

[Assignment 2 updates](#)

[27 Sep, 2010](#)

[Several items learnt \(Chapter 5\):](#)

[Assignment 2 updates](#)

[28 Sep, 2010](#)

[Several items learnt \(Chapter 5\):](#)

[My VelocityBasedAnimation class](#)
[Discussion Board > Prototype 1 – Planning](#)

[29 Sep, 2010](#)

[Discussion Board > Prototype 1 – Planning](#)

30 Sep, 2010

Assignment 2 updates - SquentialAnimation, and CardThrowingCode
Assignment 2 - Game experience

1 Oct, 2010

Assignment 2 updates - NotifcationBox

2 Oct, 2010

Assignment 2 updates - Disaster, CardDeck,Reshuffle

4 Oct, 2010

Several items learnt (Chapter 6):

Load a bitmap

Slice Bitmap

My implementation of Puzzle, PuzzlePiece, IGridItem and Grid

5 Oct, 2010

Discussion Board > Prototype 1 – Implementing and Testing

6 Oct, 2010

My implementation of DragDropTools, ISupportDrag and ISupportDrop

Discussion Board > Prototype 1 – Implementing and Testing

8 Oct, 2010

Several items learnt (Chapter 7):

The sins and cos functions

Convert radians to degrees, or vice versa

The atan functions

My implementation of AngleMoveVelocityAnimation

12 Oct, 2010

Several items learnt (Chapter 8):

My implementation of ParallelVelocityAnimation

15 Oct, 2010

Several items learnt (Chapter 8):

Match Three

Multi-Level array

Discussion Board > Prototype 1 – Implementation and Testing

18 Oct, 2010

Several items learnt (Chapter 9):

String search

String replace

Convert string to array, and vice versa

Other string functions

TextFormat

TextFormat properties

TextField properties

Samples in Chapter 9

20 Oct, 2010

Several items learnt (Chapter 9):

Samples in Chapter 9

[Label for a loop](#)

[21 Oct, 2010](#)

[Several items learnt \(Chapter 10\):](#)

[Samples in Chapter 10](#)

[Embed XML in code](#)

[Print a letter using Charcode](#)

[Generate random values \(or shuffleAnswers\)](#)

[22 Oct, 2010](#)

[Several items learnt \(Chapter 10\):](#)

[Samples in Chapter 10](#)

[25 Oct, 2010](#)

[Discussion Board > Prototype 2 – Planning](#)

[4 Nov, 2010](#)

[Assignment 3 updates -](#)

[6 Nov, 2010](#)

[Assignment 3 updates - Collision](#)

[Discussion Board > Prototype 2 – Planning](#)

[9 Nov, 2010](#)

[Discussion Board > Prototype 2 – Implementation and Testing](#)

[11 Nov, 2010](#)

[Discussion Board > Prototype 2 – Implementation and Testing](#)

[12 Nov, 2010](#)

[Discussion Board > Prototype 2 – Implementation and Testing](#)

[13 Nov, 2010](#)

[Several items learnt \(Chapter 11\):](#)

[15 Nov, 2010](#)

[Discussion Board > Prototype 2 – Implementation and Testing](#)

6 Sep, 2010

Course materials is not arrived yet, so I got a chance to look closer to the Course Outline:

There are 4 Assessment items with no exam:

- ~~Assignment 1 (Due : 9:00am 20 Sep)~~
- ~~Assignment 2 (Due : 9:00am 18 Oct)~~
- Assignment 3 (Due : 9:00am 22 Nov)
- Learning Journal (Due : 9:00am 29 Nov)

The Game prototype require either:

- Artwork, or
- Programming, or
- Both

As I am not good in artwork, I may use some public domain ones, and I will focus on the programming work, I knew Flash do provides Action Scripts 3, I think it should be powerful enough to design what I wanted.

The Game prototype should be

1. Works
2. Well documented
3. Original
4. Look and feel consistent, and lastly
5. Fun!

I have noticed an e-book version of textbook is available on [Amazon](#).

I learnt from the book Mark Pilgrim's book (#2, Ch8), that I can add a cache manifest in the html file that contained the flash file, so it can be run when offline, a cache manifest is attached in the html element, e.g. <html manifest="/cache.manifest">. The cache manifest have a header ("CACHE MANIFEST"), and below that there's 3 sections :

1. "explicit" section : default section, with a "CACHE:" header. If 2 and 3 is not used, then no section headers needed, these resource will be cached locally, I am going to put the html and flv file in this section.
2. "fallback" section : with a "FALLBACK:" header, if explicit items is failed to cache, it will use the item in fallback section, I may create a page with go online message in this section.
3. "online whitelist" section : with a "NETWORK:" header, files in this section is not cached, if I am going to add an ad on the html file I will add it here.

So the cache manifest should look like the following :

```
CACHE MANIFEST
FALLBACK:
//go-online.html
NETWORK:
*
CACHE:
/thegame.html
/thegame.flv
```

Discussion Board > 01 – My Game

I enrolled in Master of Information Technology, which didn't supply me Adobe Flash CS3/CS4, where can I obtain this software?

My game is supposed to be a two-player competitive game, but as I have no knowledge of network code there will be two separated prototypes :

Player 1 (Action)

Player 1 carry a basket, reside in the bottom of the screen. A number of objects moving from top of the screen to the bottom.

There are three kinds of objects :

1. Neutral objects - should be caught by the basket. If player 1 directly hit the object, hp is still reduced.
2. Good objects - give bonus like hp and points when hit directly.
3. Bad objects - reduce hp when hit directly.

So the key objectives is to to catch as much neutral and good objects as possible, while evading the bad objects.

Player 2 (Puzzle)

In front of player 2, there's a predefined number of cards, each represent an object but not shown.

The player have to select two cards, and match them, if succeeded, bad objects will be created on Player 1 screen, but if failed, good objects will be created instead.

The game ends when player 2 is out of cards.

7 Sep, 2010

Several items learnt (Chapter 1):

Topics

- a sprite is a frame in movie clip, which is also display objects.
- display list maintain a list of display objects

The interface

- Stage, which is main work area
- Library hold all media needed in the game.
- Timeline, which contains a number of layers (y-axis), which contains frames (x-axis), it's possible to set top layer of each frame.

ActionScript

- Smallest step approach.
- Save has to be done manually, it wont auto-save before compile.
- How to write comment, and to create procedure and functions
 - `function doSomething() { ... }`
 - `function myfunction(myNumber:Number) : Boolean { ... return true; }`
- Data Type
 - Signed integer = int,
 - Signed floating point number = Number
 - Text = String
- Methods
 - `trace()` method add debug messages in output, something like `Debug.WriteLine()` in C#
 - TextField display-object provide selectable text (which allows copy) on screen.
 - `addChild()` method add a control to the screen.
 - `removeChild()` to remove a control from screen.
 - To dispose a control, get rid of all reference and `removeChild()` is required.
 - `gotoAndStop()` change frame.
- Syntax
 - for loop = `for (var i:int=0;i<10;i++) { ... }`
 - Conditional statement = `if (myValue ==1) { ... }`
 - While loop = `while (i < 10) { ... }`
 - DoWhile loop = `do { ... } while (i < 10);`
 - Debugging, and Setting a breakpoint

Discussion Board > General Q&AS

I received my course material,

You may want to purchase e-Book version of the book, which can be found here:

<http://www.amazon.com/ActionScript-Game-Programming-University-ebook/dp/B000UZNRUA/>

Module 1 reading is Chapter 1.

Joseph

Hello

I would like to know the max word lengths of Learning Journal, because under Submission Requirements (Module 0), it states:

"With regard to the learning journal and forum discussions, word length should be between 200 and 250 words,"

I have just started writing the Learning Journal, and the word count is already more than that.

Thank you

Joseph

Discussion Board > 01 – My Game

Hello

Agreed, I am going to do it in that way, both prototypes will share the same graphics.

Do Flash support Client-Server approach? For examples, one player open hosting.html and host a server, another player open game.html and connect the server by entering the ip address of the server, and start the 1v1 game.

Thank you.

Joseph

8 Sep, 2010

Several items learnt (Chapter 2):

Functions (setChildIndex)

- setChildIndex() - which set the Z-Index of a control (e.g. setChildIndex(myWebLink, 0);) it can only be called AFTER it's added to it's parent (e.g. screen)

Buttons

Create new items in the library, by selecting Creating new Symbols, and set the linkage label by double-click on the appropriate cell in Library, I can't find the Linkage properties in my CS5 trial.

This allow us to create MovieClip and Button. Button are similar to MovieClip except it have 4 specialized frames.

(but why don't they allow us to rename frames?)

Create SimpleButton by code, it should be useful if creating more than one similar button.

```
var mySimpleButton:SimpleButton = new SimpleButton(new ButtonUp(), new ButtonOver(),
new ButtonDown(), new ButtonHit());
mySimpleButton.x = 450;
mySimpleButton.y = 250;
addChild(mySimpleButton);
```

EventListener and MouseEvent

Define event listener like the following to intercept events:

e.g.

```
myMovieClip.addEventListener(MouseEvent.CLICK, clickMascot);
function clickMascot(event:MouseEvent) { trace("mascot!"); }
```

I read the information about the MouseEvent (#3), and figure out some useful attributes like localX and localY, which shows the location inside the control, and delta, which shows the mousewheel movements.

Changing the event MouseEvent.DOUBLE_CLICK seems doesn't work in the demo, the event simply not raised.

Drawing using graphics

Every display later expose it's graphics layer through the graphics property, one can draw things on it. A numbers of methods available:

- beginFill(Color)
- lineStyle(penWidth, Color)
- moveTo(x,y)
- lineTo(x,y)
- CurveTo(midX, midY,endX,endY)
- drawRect(x,y,width, height)
- drawRoundRect(x,y,width, height, curvedCornerWidth, curvedCornerHeight)
- drawCircle(centreX, centreY, radius)
- drawEllipse(x,y, width, height)

DataType: Color is represented by hex value, e.g. 0x000000 equals to black

Sprite is a code only version of graphical control, similarly, it expose a graphics property so one can draw things on it. (#1)

```
var sprite1:Sprite = new Sprite();
sprite1.graphics.lineStyle(2,0x000000);
sprite1.graphics.beginFill(0xCCCCCC);
sprite1.graphics.drawRect(0,0,200,200);
sprite1.x = 50;
sprite1.y = 50;
addChild(sprite1);
```

More TextField

TextField has a number of useful properties like (x, y, width, height, border, selectable), it's actually possible to make it display html by using the htmlText property.

A link in htmlText can be a real uri like http:// or mailto://, or a internal event, e.g. (#1)

```
<a href='event:testing'>link</a>
```

(W3C recommend using lowercase for html tags #5)

Then in the code: (#1)

```
addEventListener(TextEvent.LINK, textLinkClick);
function textLinkClick(event:TextEvent) {
    trace(event.text == "testing"); //True displayed
}
```

Because it's html, so it's style is settable via StyleSheet (#1)

```
var myStyleSheet:StyleSheet = new StyleSheet();
myStyleSheet.setStyle("A",{textDecoration: "underline", color: "#0000FF"});
```

Further investigation to the second part of the `setStyle` function reveals it's an object. Actually it meant to create an object with a property in it's array `textDecoration` equals to "underline" and color equal to "#0000FF". So the code below does the same:

```
var style:Object = new Object(); //Object is parent of everything
style["textDecoration"] = "Underline";
style["color"] = "#0000FF";
var myStyleSheet:StyleSheet = new StyleSheet();
myStyleSheet.setStyle("A",style);
```

As for the style sheet, it's actually similar to the following in css (#6)
`a { text-decoration:underline, color:#0000FF }`

There is something like "Implicitly Typed Local Variables (var) in .Net (#2, Ch14)
e.g.

```
var patent1 = new { Title = "Bifocals", YearOfPublication = "1784"};
Debug.WriteLine(patent1.Title);
```

The difference is that the version in .Net is strongly typed (i.e. not an object), the version in Flash is an Associative Array, which store pairs of string key and object value.

To set it's font one have to use the `TextFormat` class,

```
var myFormat:TextFormat = new TextFormat();
myFormat.font = "Arial";
myText.defaultTextFormat = myFormat;
```

`TextFormat` included font, size, color, bold, italic, underline, url, target, align, leftMargin, rightMargin, indent, leading properties.(#4)

9 Sep, 2010

Several items learnt (Chapter 2):

Functions (string.fromCharCode(), getTimer())

- String.fromCharCode() - accept a char code and return it's string implementation.
- getTimer() - return how many milliseconds past after the Flash ran

Events (ENTER_FRAME, Mouse & Keyboard Events, Timer)

- Event.ENTER_FRAME - raised in every frame is loaded. The frequency of event raise can be set by setting Frame rate in properties.
- MouseEvent.ROLL_OVER - raised when mouse cursor enter a graphical control.
- MouseEvent.ROLL_OUT - raised when mouse cursor left a graphical control.
- KeyboardEvent.KEY_DOWN - raised when a key (represent by charcode) is pressed, also have property ctrlKey, shiftKey and altKey to represent if the respective key is down.

e.g. (#1)

```
stage.addEventListener(KeyboardEvent.KEY_DOWN, keyDownFunction);
function keyDownFunction(event:KeyboardEvent) {
    keyboardText.text = "Key Pressed: "+String.fromCharCode(event.charCode);
    if (event.ctrlKey)
        spacebarText.text = "Ctrl is DOWN.";
}
```

- KeyboardEvent.KEY_UP - similar to DOWN, raised when a key of released.
- TimerEvent.TIMER - raised when a timer is triggered, timer is a invisible control which when enabled, will generate events based on a interval (in milliseconds).

e.g. (#1)

```
var myTimer:Timer = new Timer(80); //80 milliseconds
myTimer.addEventListener(TimerEvent.TIMER, timerTicked);
```

Properties (MouseX/Y)

- mouseX and mouseY store current position of the mouse.

More TextField

TextField support input by setting type property to TextFieldType.INPUT, e.g. (#1)

```
var myInput:TextField = new TextField();
myInput.type = TextFieldType.INPUT;
myInput.defaultTextFormat = inputFormat;
```

```

myInput.x = 10;
myInput.y = 10;
myInput.height = 18;
myInput.width = 200;
myInput.border = true;
addChild(myInput);
stage.focus = myInput;

```

use stage object to set or get information of the flash file, one flash application have one stage only. It can set focus (focus), access the dimensions (stageWidth), get frame rate (frameRate). (#7)

Animations

There are multiple ways to create an animations

1. Using ENTER_FRAME event : e.g. (#1)

```

addEventListener(Event.ENTER_FRAME, animateHero);
function animateHero(event:Event) {
    hero.x += 7; //Move right for 7 pixels
    if (hero.currentFrame == 8) {
        hero.gotoAndStop(2);
    } else {
        hero.gotoAndStop(hero.currentFrame+1);
    }
}

```

2. Using timer and it's TIMER event :

e.g. (#1)

```

var heroTimer:Timer = new Timer(80);
heroTimer.addEventListener(TimerEvent.TIMER, animateHero);
heroTimer.start();

```

Timer has a property named currentCount, which record how many times the event is triggered.

3. Using Timer as well as getTimer() function,

e.g. (#1)

```

var gravity:Number = .00098;
var dx:Number = .2; //Velocity of the ball
var dy:Number = -.8;
var lastTime:int = getTimer(); //Time in milliseconds
function animateBall(event:Event) {
    var timeDiff:int = getTimer()-lastTime;
    lastTime += timeDiff;
}

```

```
//adjust vertical gravity by adding how much gravity generated for that amount of time (timeDiff)
    dy += gravity*timediff; //or dy = dy + (gravity * timediff)
    ball.x += timeDiff*dx;
//adjust top of the ball by adding the total vertical gravity generated.
//Why times timeDiff again?
    ball.y += timeDiff*dy;
}
```

Discussion Board > 01 – My Game

Hello

I am reading Dragging Sprites in page 67 Textbook, and decided to write a more universal code for it.

Although I think it's more elegant to create a class named Draggable and have mascot inherit from it.

How to use?

Lets say you created two mascot, mascot1 and mascot2, then you call handleDragDrop() for each mascot, e.g.

```
handleDragDrop(mascot1);
```

```
handleDragDrop(mascot2);
```

Then both mascots will support drag.

The code

...

10 Sep, 2010

Several items learnt (Chapter 2):

User Interaction

Keyboard movement is based on tracking KEY_DOWN, KEY_UP (see 9 SEP, 2010, Events) when KEY_DOWN, check and if the specific key is pressed, set the variable for that key to true, then reverse when KEY_UP. then when TIMER or ENTER_FRAME event, check if the key pressed by check that variable.

Key	Charcode
Left arrow	37
Top arrow	38
Right arrow	39
Down arrow	40
Enter	13
Space	32
Control	17
Alt	Unknown

Mouse movement is based on tracking MOUSE_DOWN, MOUSE_UP and MOUSE_MOVE, by storing start position when MOUSE_DOWN, and reset the position to default (null) when MOUSE_UP. TIMER/ENTER_FRAME event can check the movement of the mouse.

Below is a function I created for handle drag operations.

My handleDragDrop function

```
//Leung Yat Chun, 9 Sep, 2010
function handleDragDrop(mascot:Mascot)
{
    var lastPosition:Point = new Point(-1,-1);
```



```

var isDragging = false;

function onMouseDown(event:MouseEvent)
{
    if (!isDragging)
    {
        lastPosition = new Point(mouseX, mouseY);
        isDragging = true;

        stage.addEventListener(MouseEvent.MOUSE_MOVE, onMouseMove);
        stage.addEventListener(MouseEvent.MOUSE_UP, onMouseUp);
    }
}

function onMouseUp(event:MouseEvent)
{
    if (isDragging)
    {
        stage.removeEventListener(MouseEvent.MOUSE_MOVE, onMouseMove);
        stage.removeEventListener(MouseEvent.MOUSE_UP, onMouseUp);
    }
    isDragging = false;
}

function onMouseMove(event:MouseEvent)
{
    if (isDragging)
    {
        var offset:Point = new Point(mouseX-lastPosition.x, mouseY-lastPosition.y);
        mascot.x += offset.x;
        mascot.y += offset.y;
        lastPosition = new Point(mouseX, mouseY);
    }
}

mascot.addEventListener(MouseEvent.MOUSE_DOWN, onMouseDown);
}

```

Collision Detection

Collision Detection is based on the following functions in a display object:

hitTestPoint(x,y, shapeFlag) - checks if the specified location in the actual pixels of the object (if shapeFlag= true) or the bounding box (if shapeFlag= false) (#7)

hitTestObject(anotherObject) - check if the specified object hit the another object.

External data

There are various way to access external data, including :

1. External Variables

This is done by editing the flashvars parameter in the AC_FL_RunContent function in the html file, e.g. (#1)

```
... 'flashvars', 'puzzleFile=myfilename.jpg&difficultyLevel=7'
```

Then the variables can be accessed by using paramObj array, e.g.

```
var messageString = "difficultyLevel: "+paramObj["difficultyLevel"];
```

2. Loading Data

First a xml file should be created, the root tag should be <LoadingData>

```
<LoadingData><question>
    <text>Test</text>
    <answers>
        <answer type="right">Correct answer</answer>
        <answer type="wrong">Wrong answer</answer>
    </answers>
</LoadingData>
```

Then we can load it using ActionScript

```
var xmlURL:URLRequest = new URLRequest("LoadingData.xml");
var xmlLoader:URLLoader = new URLLoader();
xmlLoader.addEventListener(Event.COMPLETE, xmlLoaded);
xmlLoader.load(xmlURL); //safer this way.
```

URLRequest class represent a single HTTP request, the above means get from LoadingData.xml

URLLoader class downloads data from a URL as text, binary data, or URL-encoded variables (#7)

When xmlLoader completes it's loading, xmlLoaded is called. (#1)

```
function xmlLoaded(event:Event) {
    var dataXML = XML(event.target.data);
    trace(dataXML.question.text);
    trace(dataXML.question.answers.answer[0]);
    trace(dataXML.question.answers.answer[0].@type);
}
```

3. Saving and Loading Local Data

ActionScript have a class named SharedObject, which can be shared between different instance of the same Flash, e.g. (#1)

```
var myLocalData:SharedObject = SharedObject.getLocal("mygamedata");
```

```
trace("Found Data: "+myLocalData.data.gameinfo);
```

```
for (var prop in myLocalData.data) { //(#7)
    trace(prop+": "+myLocalData.data[prop]);
}
```

```
myLocalData.data.gameinfo = "Store this.";
```

SharedObject contains a property named data, which is a... n Object, it should be an array.

Discussion Board > 01 – My Game

Hello

Thanks, I will take a look.

Regards

Joseph Leung

13 Sep, 2010

Several items learnt (Chapter 2):

Custom Cursors

One have to create a cursor (e.g. in MovieClip), then create it and add it to the screen, then hook to the ENTER_FRAME event, which will set the cursor position to current mouseX/ mouseY.

Use Mouse.hide(); to hide the real mouse cursor, and use cursor.mouseEnabled = false; to make the custom cursors visible to mouse events.

I have created a function for custom cursors.

```
function handleCursorOp(cursor:Cursor, frameDefault:int, frameMouseDown:int)
{
    function cursorDown(event:MouseEvent)
    {
        cursor.gotoAndStop(frameMouseDown);
    }

    function cursorUp(event:MouseEvent)
    {
        cursor.gotoAndStop(frameDefault);
    }

    function updateCursor(event:Event)
    {
        cursor.x = mouseX;
        cursor.y = mouseY;

        var visibility:Boolean = true;
        if ((mouseX - (cursor.width / 2) < 0) ||
            (mouseX + (cursor.width / 2) > 550))
            visibility = false;
        if ((mouseY - (cursor.height / 2) < 0) ||
            (mouseY + (cursor.height / 2) > 400))
            visibility = false;

        cursor.visible = visibility;
    }

    Mouse.hide();
    cursor.mouseEnabled = false;
    stage.addEventListener(Event.ENTER_FRAME, updateCursor);
    stage.addEventListener(MouseEvent.MOUSE_DOWN, cursorDown);
    stage.addEventListener(MouseEvent.MOUSE_UP, cursorUp);
}
```

```
}
```

Playing Sounds

Sound class is responsible play game sound effects the source can either be imported to Library, or use the load() function. e.g. (#1)

```
var sound2:Sound = new Sound();
var req:URLRequest = new URLRequest("PlayingSounds.mp3");
sound2.load(req);
```

Once load, it can be played anytime.

```
sound2.play();
sound2.pause();
sound2.resume();
sound2.end();
```

Loading screen (and my LoadingControl)

As usual, i created a control for that

//Leung Yat Chun, 10 Sep, 2010

```
package {
    import flash.display.*;
    import flash.text.*;
    //import flash.events.Event;
    import flash.events.*
    //import flash.display.Bitmap;

    public class LoadingControl extends Sprite {
        //private var bitmap:Bitmap = null;
        private var controlWidth:int = 300;
        private var controlHeight:int = 50;
        public static var backgroundColor:Number = 0xFFFFFFFF;
        public static var borderColor:Number = 0xCCCCCC;
        public static var foregroundColor:Number = 0x000000;
        public static var borderThickness:Number = 5;

        private var PART_Progress:Sprite = new Sprite();
        private var PART_Background:Sprite = new Sprite();
        private var PART_Text:TextField = new TextField();

        public function LoadingControl():void {
            // constructor code
            //this.bitmap = bitmap;
            addEventListener(Event.ENTER_FRAME , updateFrame);
            drawBackground(PART_Background);
```

```

        addChild(PART_Background);
        addChild(PART_Progress);
        initProgressText(PART_Text);
        addChild(PART_Text);
    }

    public function updateFrame(event:Event)
    {
        drawProgress(PART_Progress);

        updateProgressText(PART_Text);
    }

    public function getPercentageCompleted():Number
    {
        var bytesLoaded:Number = this.root.loaderInfo.bytesLoaded;
        var bytesTotal:Number = this.root.loaderInfo.bytesTotal;
        return bytesLoaded/bytesTotal * 100;
    }

    function drawBackground(background:Sprite)
    {
        background.graphics.lineStyle(borderThickness,borderColor);
        background.graphics.beginFill(backgroundColor);
        background.graphics.drawRoundRect(0,0, controlWidth, controlHeight,
        borderThickness, borderThickness);
    }

    function drawProgress(progress:Sprite)
    {
        var destWidth:Number = controlWidth * getPercentageCompleted() / 100 ;

        progress.graphics.lineStyle(0,borderColor);
        progress.graphics.beginFill(backgroundColor);

        var borderThickness2 = borderThickness * 2;
        progress.graphics.drawRoundRect(borderThickness,borderThickness,
        destWidth - borderThickness2,
        controlHeight - borderThickness2,
        borderThickness2, borderThickness2);
    }

    function initProgressText(text:TextField)
    {
        //var myFormat:TextFormat = new TextFormat();
        //myFormat.font = "Georgia";
        //myFormat.bold = true;

        //myFormat.size = 24;
        //text.text = "1234567890%";
        //text.setTextFormat(myFormat);
    }

```

```

        //text.background = true;
        //text.backgroundColor = 0xFF00FF;
        text.autoSize = TextFieldAutoSize.CENTER;
    }

    function updateProgressText(text:TextField)
    {
        var percentCompleted:int = Math.round(getPercentageCompleted());
        var textColor:Number = foregroundColor;
        if (percentCompleted > 60)
            textColor = backgroundColor;
        text.textColor = textColor;

        text.htmlText = "<font size='20' name='Georgia'><b>" +
            percentCompleted.toString() + "%</b></font>";
        text.x = controlWidth / 2;
        text.y = (controlHeight - text.height) / 2;
    }
}
}

```

Functions (Random, floor, trunc)

- Math.Random() - return a value between 0 to 1
- Math.floor(), Math.trunc() - return a integer from float by closest integer(floor) or truncate the floating number (trunc)

Properties (System information, Game theft)

- stage.stageWidth, stage.stageHeight - flash width and height
- Capabilities.playerType - StandAlone (e.g. Projector), Plugin (e.g. Mozilla), ActiveX (IE) or External
- Capabilities.language - 2 letters ISO code of the language.
- Capabilities.os - OS name and version
- Capabilities.screenResolutionX/Y - user screen resolution
- Capabilities.version - Flash player name and version
- Capabilities.serverString - encode all information above to a string.
- root.loaderInfo.url - url of the flash ran on, can be used to prevent game theft.
 if (this.root.loaderInfo.url.indexOf("flashgameu.com") != -1)
 //display error.

Array

Array hold any kind of objects, adding an item to it by using the push function and access it's length function return amount of item stored.

```
var anArray:Array = new Array();
anArray.push(10);
anArray.push("abc");
trace(anArray.length); //2
trace(anArray[0] + anArray[1]); //10abc
```

Display time

```
var timePassed:int = getTimer()-startTime;
// compute minutes and seconds
var seconds:int = Math.floor((timePassed/1000)%60);
var minutes:int = Math.floor(seconds/60);
// convert to clock string
var timeString:String = minutes+":."+String(seconds+100).substr(1,2);
```


14 Sep, 2010

Several items learnt (Chapter 3):

Creating game pieces

Multiple-symbol method - create a symbol (movie clip) for each game pieces.

Single-Symbol method - create a set of playing pieces, and they share a number of common graphics like border and background.

MovieClips

Flash movie must be exported (using Export for ActionScript in Properties of MoveClip) to be usable in code. (e.g. `var c:Card = new Card();`)

Variables in dynamic (class that not defined in structure) is not required to define explicitly, they are called dynamic properties, e.g. you can set `c.cardface` without defining it.

Each frame can have a label (settable in properties \ Name), it can be used in `gotoAndXXX(frameName);`

More Array (splice) command

`array.splice(startIndex, deleteCount)` - spllice command can remove an item from list.

The array construction is a bit misleading, i would suggest to add 2 when pushing to cardlist, e.g.

`cardlist.push(i+2);` instead of changing it in `c.gotoAndStop()`, or create an additional function for that. (page 93, 98, 99) all display these misleading code.

Events

`MouseEvent.CLICK` - raise when user simple click on a card, then use

`MouseEvent.currentTarget` to return item clicked, it is an object so datatype have to be set explicitly, using the `as` keyword.

```
public function clickCard(event:MouseEvent) {
    var thisCard:Card = (event.currentTarget as Card); // what card?
    trace(thisCard.cardface);
}
```

```
}
```

if the object is not a Card, the expression will return null. (#7)

Properties

numChildren - return number of children in the specific container.

Hard code values

```
private static const boardWidth:uint = 6;
```

- private means can only be accessed inside the class. (opposite = public)
- static means same value in all instance
- const means the value can never changed.

15 Sep, 2010

Flip Animations

Flip animation of card is done by setting the animate scaleX to 0 (which takes a value from 0 to 1), then when the scaleX = 0, change the cardface, then animate scaleX to 1.

My Animation class

Animation is a class designed to hold all the animation code, this will make all future code more tidy.

How to use?

```
var animation:Animation = new Animation(this);
        animation.Flip(this.currentFrame,flipToFrame,
Animation.FLIP_HORIZONTAL);
```

And the code:

```
public class Animation
{
    private var _workObject:MovieClip = null;
    public var _animation:int = ANIMATION_NONE;

    public function Animation(workObject:MovieClip):void
    {
        _workObject = workObject;
    }

    public static var FLIP_HORIZONTAL:int = 1;
    public static var FLIP_VERTICAL:int = 2;

    private static var ANIMATION_NONE:int = 0;
    private static var ANIMATION_FLIP:int = 1;

    public function Flip(fromFrame:Object, toFrame:Object, flipDirection:int)
    {
        var flipStep:int = -5; //-5 ---> 0 ---> 5
        function doFlip(event:Event):void
        {
            flipStep++;
            if (flipDirection == FLIP_HORIZONTAL)
                _workObject.scaleX = .2 * flipStep;
            else _workObject.scaleY = .2 * flipStep;

            switch (flipStep)
            {
```

```
        case 0 : _workObject.gotoAndStop(toFrame); break;
        case 5 :
            _workObject.removeEventListener(Event.ENTER_FRAME, doFlip);
            _animation = ANIMATION_NONE;
            break;
    }
}

if (_animation == ANIMATION_NONE)
{
    _animation = ANIMATION_FLIP;
    _workObject.gotoAndStop(fromFrame);
    _workObject.addEventListener(Event.ENTER_FRAME, doFlip);
}
}
```

16 Sep, 2010

My WrapPanel class

I found it's a good idea to move the card position code in MatchingGame (#1) to separate class, allow it to be reusable in the future, thus able handle more complex situations, and the result is my WrapPanel class.

WrapPanel class is similar to the one found in WPF, except the dimension of the item is fixed, and it provides much less functionality, this is required to minimize cpu usage of the flash file.

How to use

Assume the card dimension is 50x50px, with 5px border on each side.

```
var wrapPanel:WrapPanel = new WrapPanel(50 + 5, 50 + 5, 55 * boardWidth, 55 * boardHeight);
wrapPanel.width = cardHorizontalSpacing * 5
wrapPanel.x = boardOffsetX;
wrapPanel.y = boardOffsetY;
addChild(wrapPanel);
```

then each child (card) is added to the panel

```
wrapPanel.addChild(c);
```

then when completed, do an arrange command.

```
wrapPanel.Arrange();
```

And the source code:

```
public class WrapPanel extends Sprite {
    private var _itemHeight:Number = 100;
    private var _itemWidth:Number = 100;
    private var _controlHeight:Number = 1000;
    private var _controlWidth:Number = 1000;

    public var Orientation:int = ORIENTATION_HORIZONTAL;
    //public var LeftToRight:Boolean = true;

    private static var ORIENTATION_HORIZONTAL:int = 1;
    private static var ORIENTATION_VERTICAL:int = 2;

    public function Arrange()
    {
        var itemPerCycle = 1;
        if (Orientation == ORIENTATION_HORIZONTAL)
        {
            itemPerCycle = Math.floor(_controlWidth / _itemWidth);
            this.height = Math.ceil(this.numChildren / itemPerCycle) * _itemHeight;
```

```

    }
    else
    {
        itemPerCycle = Math.floor(_controlHeight / _itemHeight);
        this.width = Math.ceil(this.numChildren / itemPerCycle) * _itemWidth;
    }
    if (itemPerCycle == 0)
        itemPerCycle = 1;

    var currentCycle = 0;
    var cycleItem = 0; //item in Current cycle

    for (var i:int = 0; i < this.numChildren; i++)
    {
        var currentObject:DisplayObject = this.getChildAt(i);

        if (Orientation == ORIENTATION_HORIZONTAL)
        {
            currentObject.x = cycleItem * _itemWidth;
            currentObject.y = currentCycle * _itemHeight;
        }
        else
        {
            currentObject.x = currentCycle * _itemWidth;
            currentObject.y = cycleItem * _itemHeight;
        }
        cycleItem++;

        trace (currentObject.x + "," + currentObject.y);

        if (cycleItem >= itemPerCycle)
        {
            cycleItem = 0;
            currentCycle++;
        }

        this.width = _controlWidth;
        this.height = _controlHeight;

        this.graphics.drawRect(0, 0, width, height);
    }
}

public function WrapPanel(itemWidth:Number, itemHeight:Number, controlWidth:Number,
controlHeight:Number) {
    // constructor code
    _itemWidth = itemWidth;
    _itemHeight = itemHeight;
    _controlWidth = controlWidth;
    _controlHeight = controlHeight;
}
}

```

17 Sep, 2010

Several items learnt (Chapter 4):

Array Functions

concat() - create a duplicate of an array.

push(newItem) - add a value to end of array and return the new size of the array.

pop - return and remove last value from the array, this treat the array like a Stack.

unshift(newItem) - insert an item in the begining of an array, and return the new size.

shift() - return and remove the first item from an array, this treat the array like a Queue.

typo in page.123, should be myArray.shift()

splice(startIndex, deleteIndex, newItems1, newItems2, ...) - remove item and add item at the startIndex in one step.

indexOf(anItem) - return location of the item, -1 if not found.

sort(compareFunction, sortOptions) - sort an array, compareFunction and sortOptions are optional,

- compareFunction is something like function(a:string, b:string) : int, if <1 then A is before B, vice versa, if 0 then A = B.
- sortOptions is one of the following values :
Array.CASEINSENSITIVE, Array.DECENDING, Array.UNIQUESORT,
Array.RETURNINDEXEDARRAY, Array.NUMERIC

Dictionary

Beside array, one can also use dictionary to store key-value pairs.

Taken from Help file (#7)

```
var dict = new Dictionary();
var obj = new Object();
var key:Object = new Object();
key.toString = function() { return "key" }

dict[key] = "Letters";
obj["key"] = "Letters";

dict[key] == "Letters"; // true
obj["key"] == "Letters"; // true
obj[key] == "Letters"; // true because key == "key" is
true b/c key.toString == "key"
dict["key"] == "Letters"; // false because "key" === key
```

```
is false  
delete dict[key]; //removes the key
```


20 Sep, 2010

Several items learnt (Chapter 4):

Properties - buttonMode, useHandCursor, tabEnabled, mouseChildren

Sprite.buttonMode - the sprite behave as a button when true (#7), which include hand cursor on mouseover and receive Click event.

Sprite.useHandCursor - whether display pointer or cursor when mouseover (#7)

Sprite.tabEnabled - enable the sprite to become tabstop (#7)

DisplayObjectContainer.mouseChildren - disable user input interactivity for their child objects (#7)

My Animation abstract class and FlipAnimation/MoveAnimation classes

The time controlling is done in Animation, when timer ticks or enterframe, it calls doFrame, which is implemented in the extended classes (FlipAnimation and MoveAnimation).

FlipAnimation only have to implements the doFrame function, which takes a number between 0 to 1, and change the WorkObject's properties based on that.

MoveAnimation, also only have to implements the doFrame function, will move the DisplayObject based on the frameRatio property.

I created this classes and I hope I can create more animations in the future (Rotate?), and chain them together (e.g. Move after Flip).

```
public class Animation
{
    public static var ANIMATION_NONE:int = 0;
    public static var ANIMATION_FLIP:int = 1;

    public static var STATE_INITED:int = 0;
    public static var STATE_RUNNING:int = 1;
    public static var STATE_COMPLETED:int = 2;

    private static var TOTAL_FRAME:int = 10;

    public var UseTimer:Boolean = false;
    public var TimeSpan:int = 300; //in ms
    public var AnimationType:int = ANIMATION_NONE;
    public var AnimationState = STATE_INITED;
    public var WorkObject:MovieClip = null;

    public function Start()
    {
        var currentFrame:int = 0;
        function onFrameEvent(event:Event)
        {
            currentFrame++;
            var frameRatio:Number = currentFrame/TOTAL_FRAME;
```

```

        if (frameRatio >= 1)
        {
            doFrame(1);
            WorkObject.removeEventListener(Event.ENTER_FRAME, nFrameEvent);
        }
        else doFrame(frameRatio);
    }

    var timer:Timer;
    var startTime:int = getTimer();
    function onTimerEvent(event:Event)
    {
        var currentTime:int = getTimer();
        var frameRatio:Number = (currentTime - startTime) / TimeSpan; //0 >= step
        if (frameRatio >= 1)
        {
            doFrame(1);
            timer.stop();
            timer.removeEventListener(TimerEvent.TIMER, onTimerEvent);
        }
        else doFrame(frameRatio);
    }

    if (AnimationState != Animation.STATE_RUNNING)
    {
        AnimationState = Animation.STATE_RUNNING;
        doFrame(0);
        if (UseTimer)
        {
            timer = new Timer(TimeSpan/TOTAL_FRAME);
            timer.addEventListener(TimerEvent.TIMER, onTimerEvent);
            timer.start();
        }
        else WorkObject.addEventListener(Event.ENTER_FRAME, onFrameEvent);
    }
}

protected function doFrame(frame:Number) //frame = a value between 0 to 1
{
}

}

public class FlipAnimation extends Animation
{
    public static var FLIP_HORIZONTAL:int = 1;
    public static var FLIP_VERTICAL:int = 2;

    public var FlipDirection:int = FLIP_HORIZONTAL;

    public var FromFrame:Object;
    public var ToFrame:Object;

    //Constructor
    public function FlipAnimation(workObject:MovieClip,fromFrame:Object, toFrame:Object):void
    {
        AnimationType = Animation.ANIMATION_FLIP;
        WorkObject = workObject;
        FromFrame = fromFrame;
        ToFrame = toFrame;
    }

    protected override function doFrame(frame:Number)
    {
        // 0 < frame < 1
        //-1 < frameScale < 1
        var frameScale = (frame - 0.5) * 2;
    }
}

```

```

        if (frameScale >= 0 && WorkObject.currentFrame != ToFrame)
            WorkObject.gotoAndStop(ToFrame);

        if (FlipDirection == FLIP_HORIZONTAL)
            WorkObject.scaleX = Math.abs(frameScale);
        else WorkObject.scaleY = Math.abs(frameScale);
    }
}

public class MoveAnimation extends Animation
{
    public var FromPosition:Point;
    public var ToPosition:Point;

    public function MoveAnimation(workObject:MovieClip,fromPosition:Point, toPosition:Point)
    {
        AnimationType = Animation.ANIMATION_MOVE;
        WorkObject = workObject;
        FromPosition = fromPosition;
        ToPosition = toPosition;
    }

    protected override function doFrame(frameRatio:Number) // 0 < frameRatio < 1
    {
        WorkObject.x = FromPosition.x + ((ToPosition.x - FromPosition.x) * frameRatio);
        WorkObject.y = FromPosition.y + ((ToPosition.y - FromPosition.y) * frameRatio);
    }
}

```

21 Sep, 2010

My Animation.StartSequential and StartParallel static function

In some case I may want to chain two animations, so StartSequential(), which run animations in an array one by one, as well as StartParallel(), which run all animations at one time, is useful for some case.

To enable StartSequential, I have to define a custom event, I found a blog post provide useful information for this (#8)

How to use?

The following code will flip the card then move it to {300,300}

```
var flip:Animation = new FlipAnimation(this, currentFrame, _cardFace);
    var move:Animation = new MoveAnimation(this, new Point(this.x, this.y), new Point(300,300)
);
Animation.StartSequential(new Array(flip, move));
```

The code

```
public static function StartSequential(animations:Array)
{
    if (animations == null || animations.length == 0) return;

    var currentAnimation:Animation = animations[0] as Animation;
    for (var i:int = 1; i < animations.length; i++)
    {
        var nextAnimation:Animation = animations[i] as Animation;
        var runNextAnimation = function()
        {
            nextAnimation.Start();
            currentAnimation.removeEventListener(Animation.COMPLETED,
runNextAnimation);
        };
        currentAnimation.addEventListener(Animation.COMPLETED, runNextAnimation);
        currentAnimation = nextAnimation;
    }
    (animations[0] as Animation).Start();
}

public static function StartParallel(animations:Array)
{
    if (animations == null || animations.length == 0) return;

    for (var i:int = 0; i < animations.length; i++)
    {
        var currentAnimation:Animation = animations[i] as Animation;
        currentAnimation.Start();
    }
}
```

My GetRelativePosition function

Because my cards are placed in a CardDeck and in a WrapPanel, previous version of MoveAnimation can only move inside the parent control, so I added a new function to find the position of all the parent controls, thus able outside it.

The trace will generate something like the following :

```
[object Card11] 0 0  
[object WrapPanel] 0 0  
[object CardDeck] 50 100  
[object MainTimeline] 0 0  
[object Stage] 0 0
```

```
function getRelativePosition(workObject:DisplayObject):Point  
{  
    //trace(workObject, workObject.x, workObject.y);  
    if (workObject.parent != null)  
    {  
        var parentOffset:Point = getRelativePosition(workObject.parent);  
        return new Point(-workObject.x + parentOffset.x, -workObject.y + parentOffset.y);  
    }  
    else return new Point(-workObject.x, -workObject.y);  
}
```

23 Sep, 2010

Several items learnt (Chapter 5):

Animation class, update

Time-Based Animation can be done using ENTER_FRAME instead of TIMER, so I modified my class so it dont use Timer to trigger an animation, but ENTER_FRAME, even if Use_Timer property is used.

Use_Timer property will enable the animation to aware of the TimeSpan, it's useful if a specific animation have to in sync with another.

With the add of Animation.StartTime variable, I can now specify when an animation start.

I have added OpacityAnimation and RotateAnimation as well.

My implementation of Random cards

I created a new class named CardDeck with a WrapPanel to hold 16 cards, so two CardDeck equals to 32 cards, each CardDeck will have a set of 16 cards. The book mentioned to create an array with random number and deduct from it. I found using the opposite approach to will be simpler to understand and code, as follows:

```
//Create an empty array to hold the ids.
var addedCards:Array = new Array();
for (var i:int = 0; i < totalCards; i++)
{
    var newCardId:int;
    //Loop till found a unoccupied ID.
    do {
        newCardId = Math.ceil(Math.random() * 16);
    } while (addedCards.indexOf(newCardId) != -1)
    //Add the id to the array, so it wont be used again.
    addedCards.push(newCardId);

    //Create the card, and add it to the panel
    var newCard:Card11 = new Card11(newCardId);
    newCard.addEventListener(MouseEvent.CLICK, onCardClicked);
    panel.addChild(newCard);
}
```

panel.Arrange();

AnimatedObject (p.158, #1) move to a direction based on "Speed". AnimatedObject will be required in Assignment 2, I am thinking how to implement this in my "game engine".


Discussion Board > General Q&As

I can drag an image to library then drag the image from library to a Keyframe, but the image on keyframe disappear when I remove the image from library.

So.. I too wonder where they came from.

24 Sep, 2010

Assignment 2 updates



My game is half playable now, with the help of a number of classes, the code of game mechanism is very very simple:

```
public function OnCardPicked(event:CardEvent)
{
    var cardDeck:CardDeck = (event.target as CardDeck);
    //Prevent continue picking cards from the same deck
    cardDeck.IsEnabled = false;

    if (_pickedCard == null)
    {
        //Assign first card
        _pickedCard = event.Card;
    }
    else
    {
        if (_pickedCard.CardFace == event.Card.CardFace)
        {
            //Match, remove the cards (uncoded)
        }
        else
        {
            //Not Match, flip back the cards
        }
    }
}
```



```
        _pickedCard.FlipBack(true, 2000, 500);
        event.Card.FlipBack(true, 2000, 500);
    }

    //Reset everything
    deck1.IsEnabled = true;
    deck2.IsEnabled = true;
    _pickedCard = null;
}
}
```

I have also created my first animation, which is a roller, which will roll the matched card to the window.

27 Sep, 2010

Several items learnt (Chapter 5):

AirRaid is an action game that use speed to control sprite movement, the movement is calculated by "Speed * timePassed / 1000"

Assignment 2 updates

The pick card sequence as follows :

1. Player select a card from the first deck (not necessary the deck in left side)
2. The card flips, and the first deck become not selectable.
3. Every card in the first deck fade, except the selected one.
4. Player select another card from second deck.
5. The card flips, and the second deck become not selectable.
6. The flash compare the cards, if matches, it does nothing special at this time.
If not match, it will flip back the cards
7. At the same time, it will unfade both decks.

All Flip, Fading are animation based, the fading effect have to be done individually to each card instead of the whole deck because that would cause the select card fades too.

```
function fade(exceptWhichCard:Card11 = null)
{
    for (var i:int = 0; i < panel.numChildren; i++)
    {
        var card:Card11 = panel.getChildAt(i) as Card11;
        if (card != exceptWhichCard)
            card.Fade();
    }
}

function unfade(exceptWhichCard:Card11= null) : Animation
{
    var anim : Animation = null;
    for (var i:int = 0; i < panel.numChildren; i++)
    {
        var card:Card11 = panel.getChildAt(i) as Card11;
        if (card != exceptWhichCard)
            anim = card.Unfade();
    }
    return anim;
}
```

As you see, the `CardDeck.unfade` function do return the `OpacityAnimation` of the last card. One issues in my assignment is that player is able to select another card while at sequence 7, when the decks are unfading, so I decided to disallow this ability till last animation is completed. Both functions are lowercase as they are intended for private use only.

28 Sep, 2010

Several items learnt (Chapter 5):

PaddleBall is another action game, it create a paddle, a ball and a number of Bricks when started, when enter frame it will fire movePaddle() and moveBall() methods, which update position of both item.

When the ball hit the paddle or boundary, it recalculate the velocityX based on paddleCurve (0.005 by default), e.g. ballIDX = (newBallX-paddle.X)*paddleCurve, and change the multiply velocityY by -1.

When the ball hit a brick, it will recalculate the velocity (like above), and remove the brick using the removeChild method, and when no bricks left, gameover.

My VelocityBasedAnimation class

Animation class is time based, as I read chapter 5 I think a velocity based animation class is needed, so I created a new class. I did considered to merge two classes, or make them inherit to each other, but that will over complicate the code at this moment.

```
public class VelocityBasedAnimation extends EventDispatcher
{
    public static const EVENT_REACHBOUNDS:String = "ReachBounds";
    public static const EVENT_STARTED:String = "Started";
    public static const EVENT_STOPPED:String = "Stopped";

    private var _stopped:Boolean = true;
    public var Velocity:Number = 1;
    public var Speed:Number = 1;//Normal speed, constant once defined

    public var MinBounds:Number = Number.MAX_VALUE;
    public var MaxBounds:Number = Number.MIN_VALUE;
    public var WorkObject:MovieClip = null;

    public function VelocityBasedAnimation(workObject:MovieClip, speed:Number = 1)
    {
        WorkObject = workObject;
        Speed = speed;
    }

    var lastTick:int = getTimer();
    function onFrameEvent(event:Event)
    {
        if (! _stopped)
        {
            var currentTime:int = getTimer();
            var timeElapsed:int = currentTime - lastTick;
            lastTick = currentTime;
            var offset:Number = Velocity * Speed * timeElapsed / 1000;

            setCurrentValue(getCurrentValue() + offset);
        }
    }
}
```

```

        if (MinBounds != Number.MAX_VALUE && getCurrentValue() < MinBounds ||
            MaxBounds != Number.MIN_VALUE && getCurrentValue() > MaxBounds)
        {
            dispatchEvent(new Event(VelocityBasedAnimation.EVENT_REACHBOUNDS));
            _stopped = true;
        }
    }

    if (_stopped)
    {
        WorkObject.removeEventListener(Event.ENTER_FRAME,onFrameEvent);
        dispatchEvent(new Event(VelocityBasedAnimation.EVENT_STOPPED));
    }
}

public function Start()
{
    if (_stopped)
    {
        _stopped = false;
        dispatchEvent(new Event(VelocityBasedAnimation.EVENT_STARTED));
        WorkObject.addEventListener(Event.ENTER_FRAME,onFrameEvent);
    }
}

public function Stop()
{
    _stopped = true;
}

protected function getCurrentValue() : Number
{
    return 0;
}

protected function setCurrentValue(newValue:Number)
{
}
}

```

Basically, after the animation is started, one can alter the Velocity to change it's direction and relative speed (e.g. -1 to go back, 2 to go next quickly), the class will calculate the changes in every frame (based on timeElapsed), and use setCurrentValue() to update the target value. The animation is stopped only if Stop() is called, or if the CurrentValue hit the bounds, in first case, EVENT_STOPPED is raised, in second case EVENT_REACHBOUNDS is raised and you can decide to stop the event or not (this part is not coded yet)

Discussion Board > Prototype 1 – Planning

My first game is a sequence of matching game then jigsaw puzzle.

My planning is to have a number of sections, each section is reside in one frame, which host a custom movieclip, so we have :

1. Main screen
2. Matching game + Jigsaw puzzle
3. Action game in assignment 2
4. Game over screen

I have no experience in graphics, so most of my work will be code based, thus I am going to rewrite the code in the game.

Based on what I read, I will need:

- a way to control animation (e.g. Flip), there are two kinds of animation identified, either based on time (e.g. Flip in 10s), or based on speed (e.g. Move to right 1px per sec)
- a way to handle dragging, and when drag failed, undo the dragging (in jigsaw puzzle)
- wire the components together to create a game

I have completed the animation class (based on time), and part of the VelocityBasedAnimation (based on speed), the matching game is mostly completed

29 Sep, 2010

Discussion Board > Prototype 1 – Planning

Thanks for reply.

It's actually almost complete the "prototype" of the matching game, still a lot of work to be done.
And you are right, action game is to be complete in assignment 3.

30 Sep, 2010

Assignment 2 updates - SequentialAnimation, and CardThrowingCode

I have added a new class named SequentialAnimation, which chains a number of animations (in an array).

I have also added code for animation that emulate card throwing:

```
private function throwCardToWindow(card:Card11, windowPos:Point, groundPos:Point)
{
    var anim1:Animation = new MoveAnimation(card, MoveAnimation.GetScreenPosition(card), windowPos,
true);
    anim1.UseTimer = true; anim1.TimeSpan = 500; anim1.StartTime = 2000;

    var anim2:FlipAnimation = new FlipAnimation(card, card.CardFace, Card11.CARD_BACK);
    anim2.FlipDirection = FlipAnimation.FLIP_VERTICAL; anim2.UseTimer = true; anim2.TimeSpan = 100;

    var anim3:Animation = new OpacityAnimation(card, 1, 0.2);
    anim3.UseTimer = true; anim3.TimeSpan = 100;

    var anim4:Animation = new MoveAnimation(card, windowPos, groundPos, true);
    anim4.UseTimer = true; anim4.TimeSpan = 500;

    var animationArray:Array = new Array(anim1, anim2, anim3, anim4);

    var seq:SequentialAnimation = new SequentialAnimation(animationArray);
    var onCompleted = function()
    {
        if (deck1.panel.contains(card))
            deck1.panel.removeChild(card);
        if (deck2.panel.contains(card))
            deck2.panel.removeChild(card);
        seq.removeListener(Animation.EVENT_COMPLETED, onCompleted);
    }

    seq.addListener(Animation.EVENT_COMPLETED, onCompleted);

    seq.Start();
}
```

It's impossible to do some kind of animation using Timeline, and without Animation class, writing something like this will produce very hard to maintain code.

Assignment 2 - Game experience

When card is matched, they will be thrown outside the window. Although I havent create the graphic for the window, as well as the sound effect for that, the animation itself looks quite fun.

1 Oct, 2010

Assignment 2 updates - NotificationBox



I have added a new class named NotificationBox, which can display some message in the middle of the screen, it contains Show() and Hide() method which allows it to Pops up and Shrink down. SizeAnimation is created to support these methods.

2 Oct, 2010

Assignment 2 updates - Disaster, CardDeck.Reshuffle

Disaster sometimes happens when a player failed to match a pairs. When it's happened, it will reshuffle all cards, animation is enabled so player can actually the cards are moving itself.

```
public function Reshuffle()
{
    for (var i:int = 0; i < panel.numChildren -1; i++)
        panel.setChildIndex(panel.getChildAt(i), Math.floor(Math.random() * panel.numChildren));
    panel.Arrange(true);
}
```

4 Oct, 2010

Several items learnt (Chapter 6):

Load a bitmap

```
(new Loader()).load(new URLRequest(uri));
```

Slice Bitmap

```
target.bitmapData(sourceBitmapData, sourceRect, targetPoint)
```

My implementation of Puzzle, PuzzlePiece, IGridItem and Grid

I have read the Sliding Puzzle code, and decided to clean up it's code. The following code load a image into puzzlepieces only, as I am not going to implement sliding puzzle.

The first class is Grid, it's similar to WrapPanel, except it allocate it's child using GetColumn() and GetRow(), which is included in the IGridItem interface.

```
public class Grid extends Sprite
{
    private var _itemHeight:Number = 100;
    private var _itemWidth:Number = 100;

    function moveObject(dispatchObj:DisplayObject, newPosition:Point, animated:Boolean)
    {
        if (animated && dispatchObj is MovieClip)
        {
            var movieClipObj = dispatchObj as MovieClip;
            var oldPos = new Point(dispatchObj.x, dispatchObj.y);
            if (newPosition != oldPos)
                Animation.Start(new MoveAnimation(movieClipObj, oldPos, newPosition, false));
        }
        else
        {
            dispatchObj.x = newPosition.x;
            dispatchObj.y = newPosition.y;
        }
    }

    public function Arrange(animated:Boolean = false)
    {
        for (var i:int = 0; i < this.numChildren; i++)
        {
            var currentObject:DisplayObject = this.getChildAt(i);
            var currentGridItem:IGridItem = currentObject as IGridItem;
            var col:int = 0;
            var row:int = 0;

            if (currentGridItem != null)
            {
                col = currentGridItem.GetColumn();
                row = currentGridItem.GetRow();
            }

            moveObject(currentObject, new Point(col * _itemWidth, row * _itemHeight),
                animated);
        }
    }
}
```

```

        this.scaleX = 1;
        this.scaleY = 1;
    }

    public function Grid(itemWidth:Number, itemHeight:Number) {
        // constructor code
        _itemWidth = itemWidth;
        _itemHeight = itemHeight;
    }
}

```

Puzzle class load image into PuzzlePieces, and place them in a Grid

```

public class Puzzle extends MovieClip
{
    private static const colCount:int = 4;
    private static const rowCount:int = 3;
    private static const pieceSpace:Number = 2;
    private static const horizOffset:Number = 50;
    private static const vertOffset:Number = 50;

    private var _panel:Grid;
    private var _image:Bitmap;

    private static function sliceBitmap(image:Bitmap, rect:Rectangle):Bitmap
    {
        var retVal:Bitmap = new Bitmap(new BitmapData(rect.width, rect.height));
        retVal.bitmapData.copyPixels(image.bitmapData, rect, new Point(0,0));
        return retVal;
    }

    private function loadImage(image:Bitmap)
    {
        _image = image;
        var pieceWidth:Number = image.width / colCount;
        var pieceHeight:Number = image.height / rowCount;

        //Construct grid
        if (_panel != null)
        {
            while (_panel.numChildren > 0)
                _panel.removeChildAt(0);
            this.removeChild(_panel);
        }
        _panel = new Grid(pieceWidth, pieceHeight);
        this.addChild(_panel);

        for (var x:int = 0; x < colCount; x++)
            for (var y:int = 0; y < rowCount; y++)
            {
                var rect:Rectangle = new Rectangle(x * pieceWidth, y * pieceHeight,
                    pieceWidth, pieceHeight);
                _panel.addChild(new PuzzlePiece(sliceBitmap(_image, rect), x, y));
            }

        _panel.Arrange(true);
    }

    private function getImage(uri:String)
    {
        var loader:Loader = new Loader();
        var onLoadingCompleted = function (event:Event)
        {
            loadImage(Bitmap(loader.content));
        }

        loader.contentLoaderInfo.addEventListener(Event.COMPLETE, onLoadingCompleted);
    }
}

```

```
        loader.load(new URLRequest(uri));
    }

    public function Puzzle(uri:String)
    {
        getImage(uri);
    }
}
```

PuzzlePiece is a piece of puzzle, it's quite simple :

```
public class PuzzlePiece extends MovieClip implements IGridItem
{
    private var _imagePiece:Bitmap;
    private var _col:int;
    private var _row:int;

    public function GetRow() : int { return _row; };
    public function GetColumn() : int { return _col; };

    public function PuzzlePiece(imagePiece:Bitmap, col:int, row:int)
    {
        _imagePiece = imagePiece;
        _col = col;
        _row = row;

        this.addChild(_imagePiece);
        this.width = _imagePiece.width;
        this.height = _imagePiece.height;
    }
}
```

5 Oct, 2010

Discussion Board > Prototype 1 – Implementing and Testing

My game is a Match cards + Jigsaw puzzle combined, the match card implementation is mostly completed, so I am moving to Jigsaw puzzle. I don't like the idea about stuffing all code in one single class, so I decided to rewrite it, and I would like to share about how I implement it:

My implementation included 4 classes/interfaces :

1. IGridItem interface - defines getColumn and getRow
2. Grid class - allocate it's child (PuzzlePiece)
2. PuzzlePiece class - represents a piece of puzzle.
3. Puzzle class - represents the whole puzzle.

The first item is IGridItem interface, interface defines a numbers of methods for it's implementer to implements, in this case, the PuzzlePiece.

Then we have Grid class, it allocates it's childs, based on it's row/col info, which can be obtained by using IGridItem interface.

You can add item to a Grid using addChild() method, but after you add them, call Arrange() method to position the item correctly.

I have comment the animation code, as it's not available in this thread.

Puzzle piece represents a piece of puzzle, I have only implemented the basic here.

The `_row` and `_col` properties allow you to set where the piece is positioned.

I suggest if you are implementing Sliding puzzle, you can listen mouse down event here, then notify the Puzzle class to slide this piece.

Puzzle class holds a Grid, which holds 1+ PuzzlePieces.

getImage() methods loads a Bitmap using Loader class.

loadImage() methods creates the Grid, then create the PuzzlePieces

sliceBitmap() methods returns a part of a Bitmap, as Bitmap.

You can create the puzzle using code:

```
private var puzzle:Puzzle = new Puzzle("slidingimage.jpg");
```

```
....
```

```
this.addChild(puzzle);
```

6 Oct, 2010

My implementation of DragDropTools, ISupportDrag and ISupportDrop

The major advantage of my implementation is that I have separate the code from the entity classes (e.g. PuzzlePieces, Mascot), so it can be reused in future.

To support Drag / Drop, the entity have to implement the ISuportDrag/ISupportDrop interface

```
public interface ISupportDrop
{
    function DragOver(dragObj:ISupportDrag) : Boolean;
    function DragLeave(dragObj:ISupportDrag);
    function Drop(dragObj:ISupportDrag) : Boolean;
}

public interface ISupportDrag
{
    function Drag() : Boolean;
}
```

Then the DragDropTools (instanced) will handle the appropriate methods once the entities are registered.

Discussion Board > Prototype 1 – Implementing and Testing

Thank you.

Attached my Jigsaw prototype, I just completed my drag n drop code.

Joseph

8 Oct, 2010

Several items learnt (Chapter 7):

The sins and cos functions

	360 (0, -1)	
180 (-1,0)	(0,0)	0 (1,0)
	90(0,1)	

It's possible to use sin and cos to find the new coordinate of entity given an angle and length traveled.

The new coordinate is equal to $((\cos(\text{angle}) * \text{length}), (\sin(\text{angle}) * \text{length}))$
e.g. 90 angle with 1 pixel length = $((\cos(90) * 1), (\sin(90)*1)) = (1,0)$

$360 \text{ degrees} = 2 * \pi \text{ radians} = 2 * 3.14 = 6.26 \text{ radians}$

Convert radians to degrees, or vice versa

$\text{radians} = 2 * \pi (\text{degrees} / 360)$
 $\text{degrees} = 360 * \text{radians} / (2 * \pi)$

The atan functions

atan function can be used to determine angles of two points, the syntax is `Math.atan2(dy, dx)`, it return values between -180 to 180.

My implementation of AngleMoveVelocityAnimation

I have added a new class to work with movement based on an angle as input, to make it work I have also modified `VelocityBasedAnimation` so it work not just with `Number` but also `Point`.

12 Oct, 2010

Several items learnt (Chapter 8):

The PointBurst class create burst effect on click, which is a combination of scale and opacity effect. I believe my Animation class is more convenient to use than this. However, I found something interesting : (Gary Rosenzweig, p.272)

```
burstSprite.removeChild(tField);
parentMC.removeChild(burstSprite);
tField = null;
burstSprite = null;
delete this;
```

The above is the cleanup procedure for the sprite, the “delete this;” it can remove the class so it can be GCed (Garbage Collected).

My implementation of ParallelVelocityAnimation

It's very resource consuming for each VelocityBasedAnimation to hook the Event.ENTER_FRAME event, to solve this problem I have create a new class to run multiple VelocityBasedAnimation together, because only one class listen to the ENTER_FRAME event, the resource consumption should be reduced, and is easier to work with.

AddAnimation and RemoveAnimation can alter the animation list at run time.

```
public class ParallelVelocityAnimation extends VelocityBasedAnimation
{
    private var _animations:Array;

    public function ParallelVelocityAnimation(owner:MovieClip,
animations:Array = null)
    {
        super(owner);
        _animations = (animations == null) ? new Array() :
animations;
    }

    public function AddAnimation(animation:VelocityBasedAnimation)
    {
        _animations.push(animation);
        animation.IsStopped = false;
    }

    public function RemoveAnimation
(animation:VelocityBasedAnimation)
    {
```

```

        var idx = _animations.indexOf(animation);
        if (idx != -1)
            _animations = _animations.splice(idx, 1);
    }

    protected override function compareValue(firstValue:Object,
secondValue:Object):int
    {
        return 0;
    }

    public override function UpdateValue(timeElapsed:int)
    {
        //trace("UpdateValue with animations =" +
_animations.length);
        for (var i:int; i < _animations.length; i++)
        {
            var currentAnimation:VelocityBasedAnimation = _animations[i];
            if (!currentAnimation.IsStopped)
            {
                currentAnimation.UpdateValue(timeElapsed);
            }
        }
    }
}

```

15 Oct, 2010

Several items learnt (Chapter 8):

Match Three

In a Match Three game, there's a grid of 8x8 of various types of game pieces, player move a game piece to get a match of three or above. Game over when no matched piece.

The most hard to understanding in this game is the Array in Array, e.g.

```
matchPattern(col, row, [[0,1]], [[0,-2],[-1,-1],[1,-1],[-1,2],[1,2],[0,3]]);
```

[[0,-2],[-1,-1],[1,-1],[-1,2],[1,2],[0,3]] is a 6 items array containing a 2 items array.

Isn't it easier to change the 2 items array to a new structure (e.g. Point)?

Multi-Level array

The declaration of multi-level array is same, e.g.

```
var grid:Array = new Array();
```

```
grid[1][2] = "aaa";
```

Discussion Board > Prototype 1 – Implementation and Testing

Your game is very interesting.

There is a bug that you may want to patch, it's also found in the MatchGame in the textbook, you can actually flip one card then flip it back without losing score.

The exploration is interesting, but I prefer to use mouse pointer to move the boat because your games are mostly mouse based (Matching game, Sliding puzzle).

I found it hard to complete the matching game in time, especially when there's a couple cards that looks so similar, you may consider to increase the time available.

Joseph

Taken from my learning journal:

TextField has a number of useful properties like (x, y, width, height, border,

selectable), it's actually possible to make it display html by using the `htmlText` property.

18 Oct, 2010

Several items learnt (Chapter 9):

String search

```
mystring.indexOf("txtToSearch");  
mystring.search("/t..T*rch/"); //using wildcards (dot means one char, * means any # of chars)  
Also accept Regular Expressions, e.g. /[A-Za-z/g <== g = match globally.
```

String replace

```
mystring.replace("abc", "cde"); //support wildcard too
```

Convert string to array, and vice versa

```
myArray = myString.split(",");  
myString = myArray.join(",");
```

Other string functions

charAt, charCodeAt, concat, fromCharCode, index, lastIndexOf, match, replace, search, splice, split, string, substr, substring, toLowerCase, toUpperCase.

TextFormat

used to specify font format in TextField.

```
var txtFormat:TextFormat = new TextFormat(fontName, size, color, bold, italic, underline, url,  
target, align);  
textField.defaultTextFormat = txtFormat;  
textField.setTextFormat(txtFormat);
```

#All characters in Courier font has the same width because it's a monospaced font.

TextFormat properties

align, blockIndent, bold, bullet, color, font, indent, italic, kerning, leading, leftMargin, letterSpacing, rightMargin, size, tabStops, target, underline, url

TextField properties

autosize, background, backgroundColor border, borderColor, defaultTextFormat, embedFonts, multiline, selectable, text, textColor, type, wordwrap

Samples in Chapter 9

Animated Text Example add each character in a string as text field to stage, then when timer ticks, it animate each text field.

Hang man add a masked (as _) text field to stage, then when keyup it check if the key pressed match the secret string, the text field unmask the matched key, by using string operation.

20 Oct, 2010

Several items learnt (Chapter 9):

Samples in Chapter 9

WordSearch try 1000 times to fit letters on the 2 axis grid, then fit the rest, which marked as *, to random letters. When player clicks on a letter, it set dragmode and draw outline when mouse over another letter, the game will check if the word match fit when mouse released. The game ends when all words matched.

Label for a loop

continue keyword accept loop name, loop name are defined in front of a loop.

e.g.

```
myloop:while (true)
{
    if (...) continue myloop;
}
```


21 Oct, 2010

Several items learnt (Chapter 10):

Samples in Chapter 10

Trivia Quiz load questions and answers from external xml file (trivia1.xml), then displaying them as Sprites (question and answersprites), judge if the answer is correct after the player selected an option, and display if he's right.

Embed XML in code

it's possible to embed XML document right in the action script code, e.g.

```
var myXML:XML =
  <parent>
    <item type="aaa">
      <value>xyz</value>
      <value>abc</value>
    </item>
  </parent>
  trace(myXML.item.attribute("type"));
  trace(myXML.item.@type);
  trace(myXML.item.value.child("*").length()); //number of nodes
  trace(myXML.item.value[0]);
```

Load external XML files

URLRequest and URLLoader are used to load external ml files.

```
var xmlLoader:URLLoader = new URLLoader(new URLRequest("xyz.xml"));
xmlLoader.addEventListener(Event.COMPLETE, xmlLoaded);
...
function (xmlLoaded(event:Event) { myXML = XML(event.target.data); }
```

Print a letter using Charcode

The charcode of A is 65, so to print letter start from, one can use String.fromCharCode(65+i);

Generate random values (or shuffleAnswers)

I believe the code look cleaner and allow code reuse if I generate random numbers in array first (seeds), then apply them in the answer, like the one I used in my Assignment, then the caller method can pop random values from the array.

```
private function generateRandomColRowList() : Array
```

```
{
    var retVal:Array = new Array();
    for (var x:int = 0; x < colCount; x++)
        for (var y:int = 0; y < rowCount; y++)
            retVal.splice(Math.floor(Math.random() * retVal.length), 0,
new Point(x, y));
    return retVal;
}
```

22 Oct, 2010

Several items learnt (Chapter 10):

Samples in Chapter 10

Deluxe Trivia Quiz added a timer, which limit player time to answer a question, the time left is displayed as a 26 frames Movieclip, score reduce for every timer tick past. The game also added "Hint" button to display hint (which will reduce potential score)

Picture Quiz, instead, use a flash file as question, so one can embed image inside the Quiz, insteading of listening to the COMPLETED event, it uses Loader class and add them directly to the stage using addChild() method.

```
questionSprite.addChild(questionLoader);
```

25 Oct, 2010

Discussion Board > Prototype 2 – Planning

The action part of the game features items dropped from a number of windows, which is caught or evaded by the player at the bottom of the screen. (Please take a look to my draft screenshot)

In the prototype 1, I have completed the MovieClip for open and close window, so I am going to place 9 Windows inside a Grid (Grid help allocates it's child items), which run the animation when a new FallingObject is spawned.

The player is controlled by VelocityAnimation(anim), e.g. when the player press Left key, it will set anim.Velocity to -1, and if the player press Shift+Right key, it will set anim.Velocity to +2. The anim.Velocity reset when player release the key.

VelocityAnimation have a property named Bounds, When ReachBounds event raised, it will restrict the possible values (by MaxVelocity / MinVelocity, which is not implemented yet).

The falling items are also controlled by similar method, but it have an additional things to handle : Collision. I have read about Collision Detection in Chapter 2, and I will need to implement it in the game, so FlyingObjects raise appropriate event (HitPlayer / HitGround / HitBasket) :

Collision Detection is based on the following functions in a display object:


hitTestPoint(x,y, shapeFlag) - checks if the specified location in the actual pixels of the object (if shapeFlag= true) or the bounding box (if shapeFlag= false)

hitTestObject(anotherObject) - check if the specified object hit the another object.

I am going to wrap everything I discussed above in the same class (ActionGame), which creates, moves (using VelocityAnimation) and free up objects, including Windows, Player and FlyingObjects.

4 Nov, 2010

Assignment 3 updates -



I have created a draft for the action element of the game.

The main MovieClip of the game is DoggleGame. Doggle contains the information element like score(uncompleted), health (uncompleted), gravity and generosity. It also contains DoggleGameFrame, which host the actual action game.

DoggleGameFrame, has a Hero, an array of FallingObjects, and a WrapPanel which contains 9 Windows, which I reused from the Puzzle game.

The hero is taken from the textbook (Rosenzweig, G), I wrapped the hero into a HorizontalController, which uses MoveVelocityAnimation to control it's movement.

The FallingObjects also uses MoveVelocityAnimation to control is "falling" from the window to the ground, and when it hit ground (EVENT_REACHBOUNDS), it will automatically remove itself.

All VelocityAnimations (Hero and FallingObjects) are run in a central "thread" ParallelVelocityAnimation, so collision check is done after each time ParallelVelocityAnimation is completed updating it's child animations. This should be less resource consuming and more accurate than checking it in another "thread" (e.g. timer)

6 Nov, 2010

Assignment 3 updates - Collision

As I mentioned earlier, collision is done after time `ParallelVelocityAnimation` updated the child's position, it uses `HitTestObject` to check if any of the `FallingObjects` hit the `HorizontalController` which contains the hero.

Because my hero, unlike normal one, is composed of two parts, the hero body and its basket, I used a special method to distinguish if the `FallingObject` hit the body or the basket.

```
function getHitTestPosition(fallObj:MovieClip) : Number
{
    if (hero.EmbeddedDisplayObject.scaleX == 1) //Face right
        return (hero.x - fallObj.x) / hero.width * -1;
    else return (hero.x - fallObj.x) / hero.width;
}
```

If the return value below 0, then the falling object hit the basket, which will lead to score increase and `Score` event raised.

If the return value above 0, then the falling object hit the body, which will lead to `Hit` event raised, which is caught by `DoggleGame` to reduce lives count.

Discussion Board > Prototype 2 – Planning

Attached the class diagram of my game.

9 Nov, 2010

Discussion Board > Prototype 2 – Implementation and Testing

After hours of worked I think the prototype of my action game is mostly completed, please try the game and tell me what you think about it.

You can start the action game by pressing the "Single Player" button. The game is about the hero (a bird who came from the wild) decided to sing for the occupants of the building, and try to make a living by collecting "gifts" from them.

Most graphics are based on the Gary Rosenzweig's textbook, as I don't know how to draw graphics. I made some modification to the Card symbols however.

The puzzle part of the game can be starting by pressing "Multi Player (One vs One)" then "Server (Puzzle)". I did planned for multi-player but it is not coded (yet).

11 Nov, 2010

Discussion Board > Prototype 2 – Implementation and Testing

Hi, very nice, I have encounter a few problems:

1. Collision detection problem, which the captain fall to ground.
2. Life lost when step on monkey (happened once only)
3. The "walls" seems climbable.
4. The "needle" is not obvious enough.

Thanks for your comment.

I will try to resize those gifts and fix the game over screen.

Your game have some nicely crafted artwork and that impressed me.

Hi,

I believe the exception is thrown from this line :

```
if (ExternalInterface.available)
```

```
ExternalInterface.call("setBackground", "#FFFFFF");
```

It change the background using Javascript.

Would you mind to tell me which Flash player are you running on? Because ExternalInterface requires Flash Player 9.

I compiled a new version without ExternalInterface, please try it.

12 Nov, 2010

Discussion Board > Prototype 2 – Implementation and Testing

You seems iron out most bugs, I can only find some minor ones

4. You can fall off platform in level 3 swamp.
5. The bottom part of the needle hurts.
6. Pressing Left or Right button when climbing ladder show incorrect animation.

Again, good work.

Hi

Yes, the submission included a html file, I guess thats the cause of the problem.

I attach the whole package this time, please see if the error goes away.

Thank you.

I have added a multi player mode for my game, to play this mode, you have to

1. start the server, copy the "fingerprint" code.
2. start the client, paste the code then press connect.

Every 4 pairs matched will increase one level of Generosity, and completing the Jigsaw puzzle will increase the Gravity level to 5.

I haven't tried this mode in multiple computers, I could use some help testing this mode.

Because running the game locally will trigger security dialog, I have uploaded my game to my isp server :

<http://home.netvigator.com/~mso6f/assignment3/ActionGame.html>

13 Nov, 2010

Several items learnt (Chapter 11):

Platform Game actually create a number of levels (gamelevel), which is large movieclip containing the initial state of the particular game level. The created hero can move inside it, and the enemies (labeled by enemy1/2/3) move automatically. The viewport (visible area on screen) is much smaller than the gamelevel, so every turn it scroll the screen to match the hero.

As most data stored in the gamelevel object, it's easier to write, but as the gamelevel is "static", it also make it hard to extend, e.g. what if I want to make a needle trap that consantly fall from above?

Assignment 3 updates - P2P



The game is completed and I have uploaded to discussion forum for further testing.

I ve also added a Player vs Player mode, which uses Adobe cirrus beta to act as middleman for the connection. I have added only the most basic elements to the multi-player mode:

1. Every 4 cards matched in the puzzle side, generosity level added in action side.
2. When puzzle side finished the jigsaw puzzle, gravity level is set to 5 in action side.
3. When action side got sufficient score, puzzle side show lost.
4. When action side lost all lives, puzzle side show win.
5. When either side disconnected, the another side win.

15 Nov, 2010

Discussion Board > Prototype 2 – Implementation and Testing

I think I understand your point, there's not enough gameplay.

Because I lack of art background so I have problem creating a complete world, with most artwork based on the textbook. Most of my work is code based, I have read the book and I have used many idea, but I "Mostly" haven't use any of it's code, there are around 45 action script file in my prototype 2.

It helps if you understand the story of my game.

The puzzle game is about an annoyed man throwing items off the street because a beautiful bird singing there, so the action game featured the bird collect the generous gift from the man.

My game is about 2 players complete with each other, please try my multiplayer prototype.

- #1 Rosenzweig, G (2008) Actionscript 3.0 Game Programming University, USA
- #2 Mark Pilgrim (2010) HTML5: Up and Running
- #3 Adobe LiveDocs (2010) Mouse Events, retrieved from - <http://livedocs.adobe.com/flex/3/langref/flash/events/MouseEvent.html>
- #4 Macromedia Flash Support Centre (1999) TextFormat
http://www.adobe.com/support/flash/action_scripts/actionscript_dictionary/actionscript_dictionary788.html
- #5 W3C (2002) XHTML 1.0 specification <http://www.w3.org/TR/xhtml1/#h-4.2>
- #6 Elizabeth Castro (2003) HTML for World Wide Web visual quickstart guide : with XHTML and CSS, 5th ed.
- #7 Adobe Community Help
- #8 Adobe LiveDocs (2010) Sprite, retrieved from - <http://www.adobe.com/livedocs/flash/9.0/ActionScriptLangRefV3/flash/display/Sprite.html>
- #9 8bitrocket (2007) Creating Custom Events In Flash AS3, retrieved from - <http://www.8bitrocket.com/2007/7/31/Creating-Custom-Events-In-Flash-AS3-ActionScript-3/>